
RICEFLOW: A MULTI-REGION, MULTI-PRODUCT, SPATIAL PARTIAL 
EQUILIBRIUM MODEL OF THE WORLD RICE ECONOMY 

1 INTRODUCTION 

A new version of the RICEFLOW model has been developed in order to address several aspects 

of the original model that made it difficult to calibrate and use for scenario analyses. The first 

RICEFLOW model (Durand-Morat and Wailes, 2003) was built following a spatial price 

equilibrium model specified by Takayama and Judge (1964). Rice consisted of three different 

varieties, namely, long-grain, medium and short grain, and aromatic rice, and three different 

processing levels, namely, paddy, brown or partially processed, and fully milled white rice. Rice 

within each category, e.g., white aromatic rice, was considered a homogeneous product (perfect 

substitution between imports and domestic production). As with all models, the Takayama and 

Judge specification presents pros and cons. Among the most salient features of this specification 

is the fact that new trade, that is, new bilateral flows, can be generated as a result of the given 

shock applied to the model. It implies that bilateral trade between regions r and j during the 

baseline period to which the model is calibrated is not required for bilateral trade between r and j 

to occur as a result of a given shock to the model. This is possible because trade flows are 

discovered as a result of an optimization problem. Arguably the most notorious disadvantage of 

this specification is its calibration to the baseline dataset. Calibration entails obtaining simulation 

results for the benchmark scenario that closely resemble what is actually observed in the market 

during the benchmark period with regard to volume and value of production, trade, and 

consumption. Obtaining an appropriate calibration with these spatial price equilibrium models 

usually entails compromises on the values of relevant parameters such as elasticities and 

transportation costs, and imposing numerous restrictions to account for political and market 

conditions affecting the benchmark data. Even after all these efforts are made, acceptable 

calibration might still be difficult to obtain. 

This new version of the spatial equilibrium model RICEFLOW adopts a different modeling 

approach, moving away from optimization and specifying the relevant behavioral equations 

according to neoclassical, marginal economic theory.  

This document provides an overview of the theory behind the new version of RICEFLOW. The 

most important differences with the previous specification described above are:  

1. Nonlinearity of demand and supply functions, and 
2. Heterogeneous products differentiated by country of origin (Armington specification) 

 

The partial equilibrium model described here is written in linearized form, namely, all equations, 

most of which are non-linear, are transformed into their linearized form1. Consequently, all 

variables in the model are represented by their percentage change rather than their nominal 

value. The linearization of non-linear equations has the main advantage of simplifying modeling 

in two primary ways, (1) by presenting equations in a simpler form that facilitates the analysis of 

the results, and (2) by greatly facilitating the calibration procedure, while still yielding the same 

results as “levels” models (Hertel, Horridge and Pearson 1991). 

This partial equilibrium model is spatial, namely, it accounts for the bilateral nature of trade. 

Other net-trade models account only for the net balance of trade without exploring the regional 

composition of trade (see, for instance, Wailes and Chavez, 2010).  

Unlike spatial price equilibrium models like that described by Takayama and Judge (1964), this 

model is not solved as a transaction cost optimization problem. It rather solves a system of first-

order conditions derived from neo-classical economic theory with regard to maximization of 

utility from consumption and profits from production activities.  

The equations are calibrated to a baseline market outcome. As a result of the calibration 

procedure, equations will be defined only for production sectors, bilateral trade flows, and 

consumers that register some level of activity in the baseline. For instance, if the dataset implies 

that producers in region a do not produce a given product z or that a the representative consumer 

in region b does not demand a given good y, then no matter the shocks applied to the model, 

these activities will not be created, namely, producers in region a will not produce z and the 

representative consumer in region b will not consume y.  The above implies that this model does 

not allow for new trade to be created, but rather only for existing trade to expand or contract as a 

result of the shocks applied to the model.  

                                                 
1 Although most CGE models are written in non-linear form, the popular multi-region, multi-sector GTAP model is 
written in linearized form. For more information on the linearization of CES and other functions, see Hertel, 
Horridge, and Pearson (1991), and Hertel (1997). 

The feature described above for production, trade, and consumption might be seen as 

undesirable. Evidence suggest that producers adapt their production strategies to take advantage 

of new opportunities, consumers change their consumption habits in response to changes in 

relative prices and tastes, and new trade is continuously  being created, for instance, as a result of 

changes in relative prices among regions, changes in transaction costs,  entrepreneurship, etc. As 

explained before, spatial price equilibrium models of the Takayama type overcome this 

limitation and allow for new activities, trade, and consumption to be created as a result of a 

shock to the model.  

RICEFLOW is written and run using the software package GEMPACK©. The basic relevant files 

needed to run the model in this environment are: 

1. The TABLO input file, which contains the equations of the model written down in a syntax 

which is very similar to ordinary algebra. TABLO files have a suffix .TAB, and are an input 

to the program TABLO. 

2. Data files containing the relevant input information to obtain a benchmark solution to the 

model (definition of sets; values of behavioral parameters; and benchmark monetary values 

representing the state of the sectoral economy), and outputs or solutions of a run. These have 

suffix .HAR for input data files, and .UPD for output data files.  

3. Command files contain the details of a simulation, including closures, shocks, starting data, 

and solution method. Command files have a suffix .CMF. 

4. Solution files are the main output from a simulation. They contain the change or percentage 

change results for all linearized variables. Solution files have a suffix .SL4. They are inputs 

to various post-simulation software programs including ViewSOL and AnalyzeGE.  

For an introduction to the GEMPACK environment, including the characteristics of GEMPACK 

files, see Harrison and Pearson (2002 a, b).  

2 DESCRIPTION OF THE RICEFLOW MODEL 

2.1 PRIMARY PRODUCTION 

The model assumes that activities/firms are competitive and can potentially produce more than 

one output/commodity. Moreover, different activities are allowed to produce the same 

commodity; for instance, we can specify an activity for lowland rice and another for upland rice, 

both potentially using different technologies and bundles of inputs, but producing the same 

commodity, say, long grain paddy rice.  

2.1.1 ACTIVITY LEVEL NEST 

Production is specified as a two-stage budgeting process. The first production stage determines 

the conditional demand functions for intermediates composite qi(a,r) and the value-added 

composite qva(a,r) (equations DEF_QI and DEF_QVA in Appendix Table 1). The derived 

demand for the value-added and intermediate composites are a function of the activity level 

qa(a,r), the technological characteristics of production, the producer price by activity2 

pap(a,r), and the composite price of value-added pva(a,r) and intermediates pi(a,r), 

respectively. The substitution effect resulting from the relative change in producer and composite 

prices is dictated by the elasticity of substitution ESUBA(a,r). The specification of this stage of 

the production process is flexible with respect to the functional form chosen, which simply 

amounts to selecting the value of ESUBA(a,r). A value of zero implies a Leontief functional 

form; a value of 1 implies a Cobb-Douglas production function; and finally any other positive 

value amounts to a Constant Elasticity of Substitution (CES) function.  

The model includes a number of technology-related exogenous variables that can be shocked 

arbitrarily as part of an experiment. Variables ava(a,r) and ain(a,r) represent augmenting 

technical changes in the productivity of the value-added and intermediates composites by activity 

and region, respectively. A positive change in ava(a,r) has two main effects: (1) at constant 

prices, it uniformly reduces the demand for factors of production; and (2) it lowers the cost of 

value-added, thus encouraging the expansion of production.  A positive change in ain(a,r) 

works similarly to a change in ava(a,r) but on the intermediates composite. The variable 

ao(a,r) stands for the output-augmenting technical change by activity and region; a shock to 

                                                 
2 By virtue of the zero-profit condition, the producer price by activity equals the unitary cost of production. 

ao(a,r) is equivalent to a Hick-neutral technical change. An increase in ao(a,r) has two main 

effects: (1) at constant prices, it uniformly reduces the demand for both the value-added and 

intermediates composites; and (2) it lowers the cost of production thus encouraging expansion of 

production.  

Figure 1. Structure of production 

 

2.1.2 VALUE-ADDED AND INTERMEDIATE NESTS 

At the bottom of the inverted tree in Figure 1 above are the value-added nest and the 

intermediates nest, in which the derived demands for factors of production qfa(f,a,r) and 

intermediate inputs qin(c,a,r) are determined (equation DEF_QFA for factors of production, 

DEF_QIN_END for endogenous intermediate inputs, and DEF_QIN_EXO for exogenous intermediate 

inputs). These derived demands are obtained from a cost-minimization problem subject to a 

Leontief, Cobb-Douglas, or CES production function, depending on the values of parameters 

ESUBVA(a,r) and ESUBIN(a,r) chosen by the user for substitution among factors in the value-

added nest and inputs in the intermediate nest, respectively. The functional form of these derived 

demand equations follows directly as a consequence of the assumption of constant returns to 

scale. The value-added composite price (cost) by activity and region pva(a,r) is estimated 

simply as a value-weighted average of the price of factors paid by the activity pfa(f,a,r) 

(equation DEF_PVA). Similarly, the unitary price (cost) of the intermediate composite by activity 

pi(a,r) is estimated simply as a value-weighted average of the price of endogenous and 

exogenous intermediate inputs paid by the activity, pq(c,r)  and pza(c,a,r), respectively 

(equation DEF_PI). For more details on the derivation of these equations, see Hertel, Horridge, 

and Pearson (1991), and Hertel (1997).  

By virtue of the zero profit condition on production, the producer price by activity and region 

(after taxes/subsidies on production are accounted for) pap(a,r)equals the cost of production by 

activity, which is in turn a function of the value-added and intermediates composite prices 

pva(a,r) and pi(a,r) and the respective shares of value added and intermediates in the total 

cost of production of activity a (the zero profit equation is introduced later as an accounting 

equation). 

The variable afe(f,a,r) represents the augmenting productivity change in factor  used by 

activity a in region r. An increase in afe(f,a,r) has three effects: (1) at constant prices, it 

reduces the demand for the specific factor of production; (2) it reduces the effective price of the 

factor thus encouraging factor substitution; and (3) it lowers the cost of value-added thus 

encouraging expansion of production. Likewise, the variable aie(c,a,r) represents the 

percentage change in the productivity of specific intermediate inputs by activity and region, and 

its effects are similar to those described above for afe(f,a,r).  

The linearized form of these derived demand equations is simple, and facilitates the 

decomposition of the changes in derived demand. The first term in these equations corresponds 

to the partial effect of the augmenting technical change variable on the derived demand of factors 

and intermediates; since these variables are declared as exogenous in the standard model, they 

can be shocked at the discretion of the user. The second term in the derived demand equations 

for factors of production and intermediate inputs corresponds to the expansionary effect, that is, 

how much the derived demand for any given factor of production or input changes as a result of 

an expansion in the output of the specific activity, relative changes in factor and input prices 

aside. Finally, the third term corresponds to the substitution effect, that is, to what extent the 

changes in intermediate demand for any given factor of production or intermediate input are 

explained by changes in their relative prices vis-à-vis the value-added composite price and the 

intermediate composite price, respectively. 

2.1.3 COMMODITY PRODUCTION BY ACTIVITY 
This model allows for each activity to produce a mix of potentially all commodities; the quantity 

of commodity c produced by activity a in region r qca(c,a,r) is determined simply as a fixed 

proportion of the activity level qa(a,r) (equation DEF_QCA). All activities in region r producing 

commodity c receive the same commodity price pc(c,r). Consequently, the average unit 

revenue or market price by activity and region pam(a,r) is estimated as a weighted average 

price of the commodities they produce (equation DEF_PAM). 

At equilibrium, the quantity of commodity c that each activity a in region r produces must equal 

the quantity of activity-specific commodity c demanded in r. In this model, both the supply and 

demand quantities are defined by the same variable qca(c,a,r). 

2.1.4 STOCKS 
The standard version of RICEFLOW accounts for changes in stocks of paddy rice by region. 

Changes in stocks can be specified in two ways. First, the nominal change3 in stock of type c 

paddy rice in region r delqk(c,r) can be specified as a fixed proportion of production of type-c 

paddy in region r. This specification is achieved by defining the slack variable 

stockslack(c,r) as exogenous and delqk(c,r) as endogenous (equation DEF_DELQK). The 

alternative specification allows for exogenous changes in stocks, and is specified by simply 

swapping stockslack(c,r) for delqk(c,r) in the closure definition (CMF file). 

Equation DEF_DELVK defines the nominal change in the value of stock of type-c paddy rice in 

region r as a function of the percentage change in the market price pc(c,r) and the change in 

the volume of stocks delqk(c,r)4.      

                                                 
3 For variables such as changes in stocks that, in levels, may go from positive to negative and vice versa, the 
percentage change specification might impose estimation problems. In these cases it is appropriate to work with 
nominal change variables rather than percentage change variables. For more information on this, see Harrison and 
Pearson (2002), and Horridge (2003).   
4 The form of equation DEF_DELVK is derived as follows: VKC = QK*PC; where VKC stands for the value of 
stock change, QK stands for the change in the volume of stocks, and PC stands for the market price of paddy rice. 

2.1.5 MARKETED COMMODITY OUTPUT 

For commodities other than paddy rice, the volume marketed qc(c,r) (both in the domestic and 

international markets) equals aggregate production across activities (equation DEF_QC_ECP). 

However, for paddy rice, the marketed volume must considered changes in stocks in addition to 

current aggregate production levels (equation DEF_QC_P).  

2.2 SUPPLY OF EXOGENOUS COMMODITIES 

The supply of exogenous commodities qz(c,r), understood as those commodities whose prices 

are determined outside the model, is specified simply to be a function of own prices pz(c,r). 

The sensibility of supply to changes in own prices is dictated by the supply elasticity 

ESUPC(c,r); shifts in the supply function can be modeled by shocking the supply intercept 

qzint(c,r), specified as an exogenous variable in the standard model (equation DEF_QZ). 

2.3 FACTOR MARKETS 

Like most partial equilibrium models, RICEFLOW treats total factor supply qfs(f,r) as 

exogenous. The user can deliberately specify the response to changes in own-factor market 

prices pf(f,r) by selecting the value of the factor supply elasticity ESUPF(f,r); furthermore, 

shifts in the factor supply curve can be specified through shock to the exogenous variable 

qfsint(f,r)(equation DEF_QFS).  

From the above we can infer that a perfectly inelastic supply function can be specified by 

changing the value of ESUPF(f,r) to zero. In the same way, if the user wants to specify a shift in 

the supply of factor f, then the shock must be applied on variable qfsint(f,r), which represents 

the percentage change in the intercept of the supply curve.  

The fix supply assumption is appropriate for most factors in short-run assessments; in the 

medium and long term, factor supply might adjust in response to market incentives, 

environmental changes, or other exogenous factors.   
                                                                                                                                                             
Expressing the identity above in change form (taking total derivative) we obtain: delvk = QK*delpc + PC*delqk; 
where delvk stands for the change in the value of stocks, delpc stands for the change in the market price of paddy 
rice, and delqk stands for the change in the volume of stocks. Since the model includes a percentage change variable 
for the market price of paddy rice, pc, rather than delpc, we redefine the above equation as follows:  
delvk = QK*(PC*pc/100) + PC * delqk. Rearranging we finally obtain: delvk = 0.01 * VKC * pc + PC * delqk. For 
more on the derivation of this equation see Horridge (2003). 
See that we need the data on the level of price pc. We arbitrarily set PC = 1 in the Coefficient statement in Table 1 
above. 

The model also differentiates factors based on their mobility. Perfectly mobile factors of 

production (set FM) can move freely among activities in the pursuit of higher returns; 

consequently, at equilibrium, the perfect mobility of factors determines a unique factor market 

price pf(f,r) for the entire market. Hence, the supply of mobile factors to specific activities is a 

response to equalize mobile factor market prices across activities.  

The situation is different for sluggish factors, whose market prices pfas(f,a,r) vary across 

activities and, consequently, so does their supply. The supply of sluggish factors across sectors 

qfsa(f,a,r) is specified as a CET function (equation DEF_QFSA). This stylized specification 

implies that without changes in relative factor prices across sectors, increases in the endowment 

of factors will be allocated uniformly across activities (expansionary effect). The substitution 

effect triggered by relative changes in pfas(f,a,r) across sectors is dictated by the elasticity of 

transformation ETRAF(f,r). 

The price index for sluggish factors pf(f,r) is estimated as a value-weighted average of market 

prices pfas(f,a,r)(equation DEF_PF). 

2.4 SOURCING OF IMPORTS 

Bilateral imports of commodity c by region s qms(c,r,s) imperfectly substitute among each 

other based on sources r. The sourcing of imports can be specified as a Leontief, Cobb-Douglas, 

or CES function depending on the value of parameter/elasticity ESUBM(c,r)(equation DEF_QMS). 

The market price of imports by commodity and region pmm(c,r) is estimated as the trade-

weighted average of region-specific import market prices pmms(c,s,r) (equation DEF_PMM). 

2.5 COMPOSITE COMMODITY PRODUCTION 

This model assumes that imports and domestic production of the same commodity are used as 

inputs for the production of a composite commodity. Maintaining separate accounts for imported 

and domestic products is a desirable specification that would allow us to disentangle the potential 

differences in intermediate and final demand for similar imported and domestically-produced 

goods. On the downside, maintaining separate commodity accounts demands that data on final 

and intermediate consumption be disaggregated into consumption of imports and domestic 

products, information not readily available in most regions.  

The volume of domestic output marketed domestically qd(c,r) (equation DEF_QD) and the 

volume of imports qm(c,r) (equation DEF_QM) of commodity c in region r  are allowed to be 

imperfect substitutes in the production of a composite commodity qq(c,r). This specification is 

commonly known as the Armington model (Armington 1969). This step can be seen as a 

production process that generates a composite commodity c using two inputs, namely domestic 

and imported commodity c, which can be imperfect substitutes and whose substitution is dictated 

by the Armington elasticity of substitution ESUBQ(c,r)(for a Leontief or Cobb-Douglas 

specification, ESUBQ(c,r) must be set equal to 0 or 1, respectively. Any other positive value will 

describe a CES production function).  

The composite commodity is later allocated into final and intermediate consumption. The 

assumption at this stage is that only fully milled rice goes into final consumption, while all paddy 

and brown rice composites are used as inputs by the milling sectors. This assumption closely 

resembles what is observed in the world rice market in general, except for some particular 

regions such as South Korea where a significant part of retail rice sales are of brown rice.  

Similar to the case of the wholesaler, it is assumed that the producer of the composite commodity 

incurs a cost of production per unit of output that is a function of input prices and their 

participation in the production process. Since the zero-profit assumption applies at this stage of 

production as well, this cost of production equals the supply price of composite commodity c 

pq(c,r).  

2.6 FINAL CONSUMPTION 

The preferences of final consumers are represented by a non-homothetic, constant difference 

elasticity (CDE) functional form. The functional form employed enters into the model in the 

estimation of the uncompensated own and cross-price elasticities and income elasticities, which 

in turn enter the final consumption equation. 

The calibration of the CDE function requires benchmark data on two parameters, namely, the 

substitution parameter SUBPAR(c,r) and the expansion or income parameter INCPAR(c,r), 

which can be obtained as outputs of a cross-entropy procedure with desirable, exogenous income 

and uncompensated own-price demand elasticities as targets (for more on this procedure, see Yu 

et al, 2004). These CDE parameters are used to estimate the substitution and income elasticities 

APE(c,g,r) and EY(c,r). Finally, the demand for final consumption of composite commodity c 

by region qdfc(c,r) (equation DEF_QDFC) adopts a general form independent of the functional 

form chosen to represent the demand system at this level, which associates the changes in retail 

prices pr(c,r) and total rice expenditures expn(r) to the respective substitution and income 

elasticities. In other words, if the user wants to change the specification of this demand system, 

the modifications will affect only the estimation of the relevant elasticities to account for the new 

functional form chosen; equation DEF_QDFC will remain unchanged. 

Fortunately, the most popular functional forms used to describe a demand system are special 

cases of the CDE function, which means that changing the demand theory at this level amounts 

to choosing the benchmark values of parameters SUBPAR(c,r) and INCPAR(c,r); no 

modifications to the model are needed. 

Table 1. Special cases of the CDE function. 
FUNCTIONAL FORM SUBPAR(c,r) INCPAR(c,r) 

Leontief 1 1 
Cobb-Douglas 0 1 
CES Same across all commodities 1 

 

2.7 POLICY VARIABLES 

The model accounts for the effect of five policy variables, namely taxes/subsidies on production 

(to(a,r)), on factors of production by activity (tfa(f,a,r)), on intermediate inputs by activity 

(tia(c,a,r)), on bilateral exports (txd(c,r,s)), and on bilateral imports (tms(c,r,s)). The 

model accounts for the power of these taxes/subsidies, defined for each tax/subsidy above as the 

ratio of (1) the market to the producer value of production for to(a,r), (2) the activity to the 

market value of derived demand for tfa(f,a,r) and tia(c,a,r), (3) the world to the border 

value of exports for txd(c,r,s), and finally (4) the market to the world value of imports for 

tms(c,r,s).  

Thus, in order to assess the economic impact of a change in, for instance, an ad-valorem 

production tax, the user must translate it into the respective percentage change in the power of 

the production tax to(a,r) and apply such a shock on to(a,r) to the model. For instance, 

assume that the benchmark dataset accounts for a 20-percent production tax (to(a,r) = 1.2), and 

we want to assess the impact of halving the activity tax to 10 percent (to(a,r) = 1.1). To 

implement this shock, the user must shock the exogenous variable to(a,r) by (1.1 – 1.2)/1.2 = -

0.0833 or -8.33%.    

Thus, the five policy variables introduced above account for all the taxation/support granted to 

the rice sector. The specification of policy variables can be further disaggregated to disentangle 

the effects of different policies grouped together in to(a,r), tms(c,r,s), and txd(c,r,s). For 

instance, the support granted through import tariffs and tariff-rate-quotas (TRQs) is aggregated 

in the value of tms(c,r,s); however, TRQs can be modeled in a mixed-complementarity 

problem, thus modeling the level of protection through TRQ endogenously. Moreover, the power 

of domestic support through deficiency payments, which is currently aggregated together with 

the power of other forms of domestic support in to(a,r), can also be disaggregated and 

specified endogenously. 

A more detailed specification of domestic support and market protection policies is granted, 

given the large set of policies employed to protect the rice sector worldwide and the different 

mechanisms through which they achieve their desired outcome. However, these specifications 

are associated with much higher complexity from a modeling point of view, since many of them 

imply the use of non-differentiable functions that require the use of more advance solution 

techniques. Thus, users are encouraged to expand the policy dimension of the model based on 

the particular applications/assessments for which it is being employed.     

2.8 PRICE LINKAGES 

The market price by activity (or, which is the same, the unitary market revenue by activity) 

pam(a,r) is linked to the producer price pap(a,r) through the power of the production tax 

to(a,r) (equation PL_ACTIVITY).  

The price for factors of production paid by activities pfa(f,a,r) is linked to the market price of 

factors (pf(f,r) and pfas(f,a,r) for mobile and sluggish factors, respectively) through the 

power of the factor tax tfa(f,a,r) (equations PL_FM and PL_FS for mobile and sluggish factors, 

respectively).  

The price of exogenous intermediate inputs paid by activities pza(c,a,r) is linked to the market 

price of exogenous inputs pz(c,r) through the power of the input tax tia(c,a,r) (equation 

PL_INPUT).  

The border price of exports of commodity c from region r to s pxbd(c,r,s) is linked to the 

market price of commodity c in exporting region r pc(c,r) through the power of the bilateral 

export margins xmrg(c,r,s) (equation PL_EXP1).; the world (f.o.b.) price of exports 

pxwd(c,r,s) is linked to bilateral border export price pxbd(c,r,s) through the power of the 

bilateral export tax txd(c,r,s) (equation PL_EXP2). 

 The same specification applies to imports. The world price of bilateral imports of good c from 

region r to s pmws(c,r,s) is linked to the market price of bilateral imports pmms(c,r,s) 

through the power of the bilateral import tax on good c tms(c,r,s) (equation PL_IMP). Finally, 

the world price of bilateral exports pxwd(c,r,s)and bilateral imports pmws(c,r,s) are linked 

through bilateral transportation costs tc(c,r,s) (equation PL_FOBCIF). 

The wholesale price of milled rice allocated to final consumption in region r pwh(c,r) is linked 

to the miller’s price of milled rice pq(c,r)  through the power of the wholesale margin 

wmrg(c,r) (equation PL_FC1). 

Finally, the retail price of milled rice allocated to final consumption in region r pr(c,r) is 

linked to the wholesale price of milled rice pwh(c,r)  through the power of the retail margin 

rmrg(c,r) (equation PL_FC2). 

2.9 SYSTEM CONSTRAINTS 

Microeconomic closures include (1) zero pure-profits conditions for activities and producers of 

composite commodities, and (2) market clearing conditions for factors of production and both 

endogenous and exogenous commodities. 

2.9.1 ZERO-PROFITS CONDITIONS 

Zero profit conditions are used to guarantee that no extra profits exist in any production activity; 

by forcing equality between costs and revenues, these conditions ensure that factors receive their 

normal rates of return. 

Equation ZP_PRODUCTION ensures that the price producers received per unit of activity pap(a,r) 

is exhausted in purchasing intermediate inputs and paying factors of production necessary for 

production. The activity level variable qa(a,r) is the free variable that clears this accounting 

condition.  

The same reasoning applies to the production of composite commodities (equation 

ZP_COMPOSITE), where the price pq(c,r) is fully exhausted in the purchase of production inputs, 

namely domestic and imported commodities. The production of composite commodity qq(c,r) 

is the free variable that clears this zero-profit condition. 

Zero profit conditions are reasonable assumptions for long-run assessments, since it is assumed 

that the limitations to adjust production techniques and resource allocation are less stringent in 

the long run, which would allow for factors of production to move across sectors in the pursuit of 

higher returns. This mobilization of resources would result in the elimination of extra profits in 

any particular sector. In the short run, however, the zero profit assumption might be misleading, 

ignoring the impact of barriers to entry or exit in specific sectors. 

2.9.2 CLEARING IN FACTOR MARKETS 
Equality of factor demand and supply must prevail at equilibrium. For mobile factors without 

sector-specific supply, this is enforced by equating total factor supply qfs(f,r) with the sum of 

sector-specific demands qfa(f,a,r) (equation MKTCLR_FM). This clearing condition is enforced 

through changes in mobile factor prices pf(f,r).  

For sluggish factors, factor supply by region qfsa(f,a,r) is factor and activity-specific, and 

consequently the market clearing condition is also enforced on those dimensions (equation 

MKTCLR_FS). The activity-specific factor price pfa(f,a,r) is the free variable that adjusts so as 

to clear the market for sluggish factors. 

2.9.3 CLEARING IN DOMESTIC COMMODITY MARKETS 

Equation MKTCLR_DOMCOM ensures the equality between total production of commodity c 

qc(c,r) and total demand of domestic commodity c, which comprises domestic demand 

qd(c,r) and export demands qms(c,r). The price of domestic commodity pc(c,r) is the free 

variable that adjust so as to clears this equilibrium.   

2.9.4 CLEARING IN COMPOSITE COMMODITY MARKETS 

As previously explained, the production of composite commodity c qq(c,r) utilizes domestic 

goods and imports as inputs, and are consumed as final products qdfc(c,r), and as intermediate 

inputs by the domestic production sectors qin(c,a,r). The market clearing equation for 

composite commodities subject to final consumption (equation MKTCLR_FC) ensures that the 

total demand qdfc(c,r)and total supply qq(c,r)are equal at equilibrium. The composite 

commodity price pq(c,r) is the free variable that adjusts so as to clear the market for activity-

specific commodities. For composite commodities allocated to intermediate consumption, the 

market clearing equation MKTCLR_INTERM ensures that the sum of intermediate demand by 

activities qin(c,a,r) equals the total supply qq(c,r).   

2.9.5 CLEARING IN EXOGENOUS COMMODITY MARKETS 
Exogenous commodities are demanded only as intermediate inputs; therefore, the clearing 

market equation MKTCLR_EXOCOM ensures that the sum of intermediate demand by activities 

qin(c,a,r) equals the total supply qz(c,r). In the standard closure, supply of exogenous 

commodities is specified as exogenous whereas the market price for exogenous commodities 

pz(c,r) is specified as endogenous and is the free variable that adjusts so as to balance this 

equation. Alternatively, the user can change the closure, making pz(c,r) exogenous and 

endogenizing qz(c,r). This alternative closure can be achieved simply by swapping pz(c,r) 

for excomslack(c,r). 

2.10 SECONDARY ESTIMATIONS 

The equations listed in this section are aimed at facilitating the analysis of the results, and in no 

way alter the equilibrium solution reached as a result of a given run.  

There are four main blocks of equations, namely (1) production aggregates, (2) consumption 

aggregates; and (3) trade aggregates. Equations in each block are aimed at estimating useful 

aggregations of production, consumption, and trade patterns across regions and commodities.  

4 APPENDIX 

 

Appendix Table 1. TABLO file containing the specification of RICEFLOW 

!==============================================================================! 

!                  1. INTRODUCTION OF RELEVANT INPUT DATA FILES                ! 

!==============================================================================! 

FILE BASEDATA # file containing benchmark information for initial calibration#; 

FILE(new, text) OUTFILE # updated value of coefficients #; 

!==============================================================================! 

!                           2. DEFINITION OF SETS                              ! 

!==============================================================================! 

SET ACT    # activities # maximum size 20  

           READ ELEMENTS FROM FILE BASEDATA HEADER "S01";  

SET ACT_P  # primary activities # (A_LGP,A_MGP,A_FRP); 

           SUBSET ACT_P IS SUBSET OF ACT; 

SET COM    # commodities # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S02";  

SET COM_E  # commodities with endogenous production # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S03";  

           SUBSET COM_E IS SUBSET OF COM; 

SET COM_CE # commodities with exogenous production # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S04";  

           SUBSET COM_CE IS SUBSET OF COM; 

SET COM_FC # commodities suitable for final consumption # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S05";  

           SUBSET COM_FC IS SUBSET OF COM; 

           SUBSET COM_FC IS SUBSET OF COM_E; 

SET COM_ECFC  # endogenous comm. complement of FC # = COM_E - COM_FC; 

SET COM_CFC # commodities complement of FC # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S06";  

           SUBSET COM_CFC IS SUBSET OF COM; 

           SUBSET COM_ECFC IS SUBSET OF COM_CFC; 

           SUBSET COM_CE IS SUBSET OF COM_CFC; 

SET COM_P # primary commodities # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S07";  

           SUBSET COM_P IS SUBSET OF COM; 

           SUBSET COM_P IS SUBSET OF COM_E; 

           SUBSET COM_P IS SUBSET OF COM_ECFC; 

SET COM_ECP # endogenous comm. complement of P # = COM_E - COM_P; 

Set FAC    # factors of production # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S08";  

    FM     # mobile factors of production # (labor); 

           SUBSET FM IS SUBSET OF FAC; 

SET FS     # sluggish factors of production # = FAC - FM; 

 

SET REG    # regions or foreign institutions # maximum size 20 

           READ ELEMENTS FROM FILE BASEDATA HEADER "S09"; 

!==============================================================================! 

!                       3. READING OF RELEVANT COEFFICIENTS                    ! 

!==============================================================================! 

Coefficient (parameter)(all,a,ACT)(all,r,REG)               ESUBA(a,r) 

# Elast. of substitution between value-added and intermediate composites #; 

Coefficient (parameter)(all,a,ACT)(all,r,REG)               ESUBVA(a,r) 

# Elast. of substitution among factors used by activity a in region r #; 

Coefficient (parameter)(all,a,ACT)(all,r,REG)               ESUBIN(a,r) 

# Elast. of substitution among intermediates used by activity a in region r #; 

Coefficient(parameter) (all,c,COM_CE)(all,r,REG)            ESUPC(c,r) 

# own-price supply elasticity for exogenous commodity c #; 

Coefficient (parameter)(all,c,COM_E)(all,r,REG)             ESUBM(c,r) 

# Armington elasticity of substitution for bilateral imports of commodity c #; 

Coefficient (parameter)(all,c,COM_E)(all,r,REG)             ESUBQ(c,r) 

# Elast. of subst. in the production of composite commodity c in region r #; 

Coefficient(parameter) (all,f,FAC)(all,r,REG)               ESUPF(f,r) 

   # own-price supply elasticity for factor f #; 

Coefficient (parameter)(all,f,FAC)(all,r,REG)               ETRAF(f,r) 

# Elast. of transformation for factor of production f in region r #; 

Coefficient (parameter)(all,c,COM_FC)(all,r,REG)            INCPAR(c,r) 

   # CDE income parameter by commodity and region #; 

Coefficient (parameter)(all,c,COM_FC)(all,r,REG)            SUBPAR(c,r) 

   # CDE substitution parameter by commodity and region #; 

 

Read    ESUBA   from file BASEDATA header "P01"; 

Read    ESUBVA  from file BASEDATA header "P02"; 

Read    ESUBIN  from file BASEDATA header "P03"; 

Read    ESUPC   from file BASEDATA header "P04"; 

Read    ESUBM   from file BASEDATA header "P05"; 

Read    ESUBQ   from file BASEDATA header "P06"; 

Read    ESUPF   from file BASEDATA header "P07"; 

Read    ETRAF   from file BASEDATA header "P08"; 

Read    INCPAR  from file BASEDATA header "P09"; 

Read    SUBPAR  from file BASEDATA header "P10"; 

Coefficient (ge 0)(all,c,COM_E)(all,a,ACT)(all,r,REG)     VCAM(c,a,r)  

   # Market value of commodity c produced by activity a in region r #; 

Coefficient (ge 0)(all,a,ACT)(all,f,FAC)(all,r,REG)     VEA(f,a,r)  

   # Value of demand of endowment f by activity and region #; 

Coefficient (ge 0)(all,a,ACT)(all,c,COM_CFC)(all,r,REG)     VICA(c,a,r)  

   # Value of intermediate demand of commodity c by activity and region #; 

Coefficient (all,c,COM_P)(all,r,REG)    VKC(c,r)  

   # Value of change in stock by commodity and region #; 

Coefficient (ge 0)(all,c,COM_E)(all,r,REG)(all,s,REG)     VBXM(c,r,s)  

   # Value of bilateral exports of commodity c at market price #; 

Coefficient (ge 0)(all,c,COM_E)(all,r,REG)(all,s,REG)     VBXB(c,r,s)  

   # Value of bilateral exports of commodity c at border price #; 

Coefficient (ge 0)(all,c,COM_E)(all,r,REG)(all,s,REG)  VBXW(c,r,s)  

   # Value of bilateral exports of commodity c at world price #; 

Coefficient (ge 0)(all,c,COM_E)(all,r,REG)(all,s,REG)  VBYW(c,r,s) 

   # Value of bilateral import of commodity c at world price #; 

Coefficient (ge 0)(all,c,COM_E)(all,r,REG)(all,s,REG)  VBYM(c,r,s) 

   # Value of bilateral imports of commodity c at market price #; 

Coefficient (ge 0)(all,c,COM_FC)(all,r,REG)    VFC(c,r)  

   # Bulk value of final consumption by commodity and region #; 

Coefficient (ge 0)(all,c,COM_FC)(all,r,REG)    VFCW(c,r)  

   # Wholesale value of final consumption by commodity and region #; 

Coefficient (ge 0)(all,c,COM_FC)(all,r,REG)    VFCR(c,r)  

   # Retail value of final consumption by commodity and region #; 

Coefficient (ge 0)(all,a,ACT)(all,f,FAC)(all,r,REG)     VEAM(f,a,r)  

   # Market value of demand of endowment f by activity and region #; 

Coefficient (ge 0)(all,a,ACT)(all,c,COM_CFC)(all,r,REG)     VICAM(c,a,r)  

   # Market value of interm demand of commodity c by activity and region #; 

 

Read    VCAM   from file BASEDATA header "V01"; 

Read    VEA    from file BASEDATA header "V02"; 

Read    VEAM   from file BASEDATA header "V03"; 

Read    VICA   from file BASEDATA header "V04"; 

Read    VICAM  from file BASEDATA header "V05"; 

Read    VBXM   from file BASEDATA header "V06"; 

Read    VBXB   from file BASEDATA header "V07"; 

Read    VBXW   from file BASEDATA header "V08"; 

Read    VBYM   from file BASEDATA header "V09"; 

Read    VBYW   from file BASEDATA header "V10"; 

Read    VFC    from file BASEDATA header "V11"; 

Read    VFCW   from file BASEDATA header "V12"; 

Read    VFCR   from file BASEDATA header "V13"; 

Read    VKC    from file BASEDATA header "V14"; 

!==============================================================================! 

!                     4. DEFINITION OF RELEVANT VARIABLES                      ! 

!==============================================================================! 

!--------------------------------------!                                         

!       4.1 QUANTITY VARIABLES         ! 

!--------------------------------------! 

Variable  

(all,a,ACT)(all,r,REG)                    qa(a,r)      

# Level of activity a in region r #; 

(all,a,ACT)(all,r,REG)                    qva(a,r)     

# Derived demand for the value-added composite by activity and region #; 

(all,a,ACT)(all,f,FAC)(all,r,REG)         qfa(f,a,r) 

# Derived demand for specific factors of production by activity and region #; 

(all,a,ACT)(all,r,REG)                    qi(a,r)   

# Derived demand for the intermediate composite by activity and region #; 

(all,c,COM_CFC)(all,a,ACT)(all,r,REG)     qin(c,a,r)  

# Derived demand for specific intermediate inputs by activity and region #; 

(all,a,ACT)(all,c,COM_E)(all,r,REG)       qca(c,a,r)  

# Production of specific commodities by activity and region #; 

(all,c,COM_E)(all,r,REG)                  qc(c,r)  

# Volume of commercialization by commodity and region #; 

(change)(all,c,COM_P)(all,r,REG)          delqk(c,r)  

# Change in the volume of stock by commodity and region #; 

(all,c,COM_E)(all,r,REG)                  qd(c,r)      

# Volume commercialized domestically by commodity and region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)      qms(c,r,s)      

# Bilateral volume of imports by commodity #; 

(all,c,COM_E)(all,r,REG)                  qm(c,r)      

# Volume of imports by commodity and region #; 

(all,c,COM_E)(all,r,REG)                  qq(c,r)      

# Production by composite commodity and region #; 

(all,c,COM_CE)(all,r,REG)                 qz(c,r) 

# Supply by exogenous commodity and region #; 

(all,c,COM_FC)(all,r,REG)                 qdfc(c,r)      

# Demand for final consumption by composite commodity and region #; 

(all,f,FAC)(all,r,REG)                    qfs(f,r) 

# Total supply by factor of production and region #; 

(all,a,ACT)(all,f,FS)(all,r,REG)          qfsa(f,a,r) 

# Sectoral supply by sluggish factor of production and region #; 

!--------------------------------------!                                         

!         4.2 PRICE VARIABLES          ! 

!--------------------------------------! 

(all,a,ACT)(all,r,REG)                    pam(a,r) 

# Unitary market revenue by activity and region #; 

(all,a,ACT)(all,r,REG)                    pap(a,r)  

# Unitary production cost by activity and region #; 

(all,a,ACT)(all,r,REG)                    pva(a,r) 

# Unitary cost of value-added composite by activity and region #; 

(all,f,FAC)(all,a,ACT)(all,r,REG)         pfa(f,a,r)  

# Price of factors of production by acivity and region #; 

(all,f,FAC)(all,r,REG)                    pf(f,r)  

# Market price of factors of production by region #; 

(all,f,FS)(all,a,ACT)(all,r,REG)          pfas(f,a,r)  

# Price of sluggish factors of production by activity and region #; 

(all,a,ACT)(all,r,REG)                    pi(a,r) 

# Unitary cost of intermediate composite by activity and region #; 

(all,c,COM_E)(all,r,REG)                  pc(c,r) 

# Price of domestically-produced commodities by region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       pxbd(c,r,s)  

# Border price of bilateral exports by commodity #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       pxwd(c,r,s)  

# World price of bilateral exports by commodity #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       pmws(c,r,s)  

# World price of bilateral imports by commodity #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       pmms(c,r,s)  

# Market price of bilateral imports by commodity #; 

(all,c,COM_E)(all,r,REG)                  pmm(c,r)  

# Market price of imports by commodity and region #; 

(all,c,COM_E)(all,r,REG)                  pq(c,r)  

# Price of composite commodities by region #; 

(all,c,COM_FC)(all,r,REG)                 pwh(c,r)  

# Wholesale price of final consumption commodities by region #; 

(all,c,COM_FC)(all,r,REG)                 pr(c,r)  

# Retail price of final consumption commodities by region #; 

(all,c,COM_CE)(all,a,ACT)(all,r,REG)      pza(c,a,r)  

# Price of exogenous commodities by activity and region #; 

(all,c,COM_CE)(all,r,REG)                 pz(c,r)  

# Market price of exogenous commodities by region #; 

!--------------------------------------!                                         

!       4.3 TECHNOLOGY VARIABLES       ! 

!--------------------------------------! 

(all,a,ACT)(all,r,REG)                    ao(a,r) 

# Augmenting technical change of output by activity and region #; 

(all,a,ACT)(all,r,REG)                    ava(a,r) 

# Augmenting technical change for value-added composite by activity and region#; 

(all,a,ACT)(all,r,REG)                    ain(a,r) 

# Augmenting technical change for interm. composite by activity and region #; 

(all,f,FAC)(all,a,ACT)(all,r,REG)         afe(f,a,r) 

# Augmenting technological change for primary factor f by activity and region #; 

(all,c,COM_CFC)(all,a,ACT)(all,r,REG)     aie(c,a,r) 

# Augmenting technical change for intermediate comm c by activity and region #; 

!--------------------------------------!                                         

!        4.4 POLICY VARIABLES          ! 

!--------------------------------------! 

(all,a,ACT)(all,r,REG)                    to(a,r)  

# Production tax by activity and region #; 

(all,f,FAC)(all,a,ACT)(all,r,REG)         tfa(f,a,r)  

# Tax/subsidy on factor f by activity and region #; 

(all,c,COM_CE)(all,a,ACT)(all,r,REG)      tia(c,a,r)  

# Tax/subsidy on input c by activity and region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       txd(c,r,s)  

# Bilateral export tax by commodity #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       tms(c,r,s)  

# Bilateral import tariffs by commodity #; 

 

!--------------------------------------!                                         

!          4.5 OTHER VARIABLES         ! 

!--------------------------------------! 

(change)(all,c,COM_P)(all,r,REG)          delvk(c,r)  

# Change in the value of stock by commodity and region #; 

(change)(all,c,COM_P)(all,r,REG)          stockslack(c,r)  

# Slack variable for stocks by commodity and region #; 

(all,a,ACT)(all,r,REG)                    zpactslack(a,r)  

# Slack variable for zero profits by activity and region #; 

(all,c,COM_CE)(all,r,REG)                 excomslack(c,r)  

# Slack variable for market clearing by exogenous commodity and region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       qmsint(c,r,s) 

# Shift in bilateral import demand intercept by commodity #; 

(all,f,FAC)(all,r,REG)                    qfsint(f,r) 

# Shift in supply intercept by factor and region #; 

(all,c,COM_CE)(all,r,REG)                 qzint(c,r) 

# Shift in supply intercept by exogenous commodity and region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       xmrg(c,r,s) 

# Bilateral export margin by commodity and region #; 

(all,c,COM_FC)(all,r,REG)                 wmrg(c,r)  

# Wholesale margin by final consumption commodities and region #; 

(all,c,COM_FC)(all,r,REG)                 rmrg(c,r)  

# Retail margin by final consumption commodities by region #; 

(all,c,COM_E)(all,r,REG)(all,s,REG)       tc(c,r,s)  

# Bilateral transportation cost by commodity and region #; 

(all,r,REG)                               expn(r) 

# Expenditure on final consumption by region #; 

(all,r,REG)                               pop(r) 

# Rate of population growth by region #; 

 

!==============================================================================! 

!                              5. DATA UPDATES                                 ! 

!==============================================================================! 

   Update (all,c,COM_E)(all,a,ACT)(all,r,REG) 

   VCAM(c,a,r) = pc(c,r) * qca(c,a,r); 

   Update (all,a,ACT)(all,f,FAC)(all,r,REG) 

   VEA(f,a,r) = pfa(f,a,r) * qfa(f,a,r); 

   Update (all,a,ACT)(all,c,COM_ECFC)(all,r,REG) 

   VICA(c,a,r) = pq(c,r) * qin(c,a,r); 

   Update (all,a,ACT)(all,c,COM_CE)(all,r,REG) 

   VICA(c,a,r) = pza(c,a,r) * qin(c,a,r); 

   Update (change)(all,c,COM_P)(all,r,REG) 

   VKC(c,r) = delvk(c,r); 

   Update (all,c,COM_E)(all,r,REG)(all,s,REG) 

   VBXM(c,r,s) = pc(c,r) * qms(c,r,s); 

   Update (all,c,COM_E)(all,r,REG)(all,s,REG) 

   VBXB(c,r,s) = pxbd(c,r,s) * qms(c,r,s); 

   Update (all,c,COM_E)(all,r,REG)(all,s,REG) 

   VBXW(c,r,s) = pxwd(c,r,s) * qms(c,r,s); 

   Update (all,c,COM_E)(all,r,REG)(all,s,REG) 

   VBYW(c,r,s) = pmws(c,r,s)* qms(c,r,s); 

   Update (all,c,COM_E)(all,r,REG)(all,s,REG) 

   VBYM(c,r,s) = pmms(c,r,s)* qms(c,r,s); 

   Update (all,c,COM_FC)(all,r,REG) 

   VFC(c,r) = pq(c,r) * qdfc(c,r); 

   Update (all,c,COM_FC)(all,r,REG) 

   VFCW(c,r) = pwh(c,r) * qdfc(c,r); 

   Update (all,c,COM_FC)(all,r,REG) 

   VFCR(c,r) = pr(c,r) * qdfc(c,r); 

   Update (all,a,ACT)(all,f,FAC)(all,r,REG) 

   VEAM(f,a,r) = pf(f,r) * qfa(f,a,r); 

   Update (all,a,ACT)(all,c,COM_ECFC)(all,r,REG) 

   VICAM(c,a,r) = pq(c,r) * qin(c,a,r); 

   Update (all,a,ACT)(all,c,COM_CE)(all,r,REG) 

   VICAM(c,a,r) = pz(c,r) * qin(c,a,r); 

 

Coefficient       TINY 

   # Small number to prevent zerodivides or singular matrix #; 

Formula TINY = 0.0000000001*RANDOM(0,1); 

 

!==============================================================================! 

!                           6. BEHAVIORAL EQUATIONS                            ! 

!==============================================================================! 

!--------------------------------------! 

!            6.1 PRODUCTION            ! 

!--------------------------------------! 

!--------------------------------------! 

!      6.1.1 ACTIVITY LEVEL NEST       ! 

!--------------------------------------! 

Equation DEF_QVA     

   # Derived demand for the value-added composite by activity and region # 

   (all,a,ACT)(all,r,REG)  

   qva(a,r) = qa(a,r) - ava(a,r) - ao(a,r) +  

                    ESUBA(a,r) * {pap(a,r) + ava(a,r) + ao(a,r) - pva(a,r)}; 

   !effective quantity of value-aded composite = [qva(a,r) + ava(a,r)]! 

   !At a given activity level, an increase in the efficiency in production or in 

    the use of value-added (Hicks-neutral wrt factors of production) will result 

    in a decrease in the derived demand for the value-added composite 

    and viceversa! 

 

Equation DEF_QI     

   # Derived demand for the intermediate composite by activity and region # 

   (all,a,ACT)(all,r,REG)  

   qi(a,r) = qa(a,r) - ain(a,r) - ao(a,r) + 

                    ESUBA(a,r) * {pap(a,r) + ain(a,r) + ao(a,r) - pi(a,r)}; 

   !effective quantity of intermediate composite = [qinta(a,r) + ain(a,r)]! 

   !At a given activity level, an increase in the efficiency in production or in 

    the use of intermediates (Hicks-neutral wrt intermediates) will result in a 

    decrease in the derived demand for the intermediate composite and viceversa! 

!--------------------------------------! 

!        6.1.2 VALUE-ADDED NEST        ! 

!--------------------------------------! 

Equation DEF_QFA     

   # Derived demand for factors of production by activity and region # 

   (all,f,FAC)(all,a,ACT)(all,r,REG)  

   qfa(f,a,r) = -afe(f,a,r) + qva(a,r) +  

            ESUBVA(a,r) * {pva(a,r) - [pfa(f,a,r) - afe(f,a,r)]}; 

   !effective quantity of factor f by activity a = [qfa(f,a,r) + afe(f,a,r)]! 

   !At a given demand for value-added composite, an increase in the efficiency 

    in the use of a specific mobile factor of production will result in a  

    decrease in the derived demand for the mobile factor of production 

    and viceversa! 

 

zerodivide default 0.3333; 

Coefficient (all,f,FAC)(all,a,ACT)(all,r,REG)     SFVA(f,a,r) 

   # Share of factors in value-added cost by activity and region #; 

   Formula (all,f,FAC)(all,a,ACT)(all,r,REG) 

   SFVA(f,a,r) = VEA(f,a,r) / sum{g,FAC, VEA(g,a,r)} ; 

   Write SFVA to file OUTFILE; 

zerodivide off; 

 

Equation DEF_PVA     

   # Unitary cost of value-added composite by activity and region # 

   (all,a,ACT)(all,r,REG)  

   pva(a,r) = sum{f,FAC, SFVA(f,a,r) * [pfa(f,a,r)-afe(f,a,r)]}; 

   !effective price of factor f by activity = [pf(f,r) - afe(f,a,r)]! 

   !This equation can also be understood as a zero profit condition in the  

    production of value-added! 

 

!--------------------------------------! 

!       6.1.3 INTERMEDIATES NEST       ! 

!--------------------------------------! 

Equation DEF_QIN_END     

   # Derived demand for endogenous intermediate inputs by activity and region # 

   (all,c,COM_ECFC)(all,a,ACT)(all,r,REG)  

   qin(c,a,r) = -aie(c,a,r) + qi(a,r) +  

              ESUBIN(a,r) * {pi(a,r)-[pq(c,r)-aie(c,a,r)]}; 

Equation DEF_QIN_EXO     

   # Derived demand for exogenous intermediate inputs by activity and region # 

   (all,c,COM_CE)(all,a,ACT)(all,r,REG)  

   qin(c,a,r) = -aie(c,a,r) + qi(a,r) +  

              ESUBIN(a,r) * {pi(a,r)-[pza(c,a,r)-aie(c,a,r)]}; 

   !effective quantity input c by activity and region=[qint(c,a,r)+aie(c,a,r)]! 

   !At a given demand for the intermediate composite, an increase in the  

    efficiency in the use of a specific input will result in a decrease in the 

    derived demand for that input and viceversa! 

 

zerodivide default 0.0909;  

Coefficient (all,c,COM_CFC)(all,a,ACT)(all,r,REG)     SIIN(c,a,r) 

   # Share of inputs in intermediates cost by activity and region #; 

   Formula (all,c,COM_CFC)(all,a,ACT)(all,r,REG) 

   SIIN(c,a,r) = VICA(c,a,r) / sum{p,COM_CFC, VICA(p,a,r)} ; 

   Write SIIN to file OUTFILE; 

zerodivide off; 

 

Equation DEF_PI     

   # Unitary cost of intermediate composite by activity and region # 

   (all,a,ACT)(all,r,REG)  

   pi(a,r) = sum{c,COM_ECFC, SIIN(c,a,r) * [pq(c,r) - aie(c,a,r)]} +  

             sum{c,COM_CE, SIIN(c,a,r) * [pza(c,a,r) - aie(c,a,r)]}; 

 !effective price input c in set COM_E by activity = [pq(c,r) - aie(c,a,r)]! 

 !effective price input c in set COM_CE by activity = [pza(c,a,r) - aie(c,a,r)]! 

 !This equation can also be understood as a zero profit condition in the  

  production of value-added! 

 

!--------------------------------------! 

!     6.1.4 COMMODITY BY ACTIVITY      ! 

!--------------------------------------! 

Equation DEF_QCA     

   # Production of specific commodities by activity and region # 

   (all,a,ACT)(all,c,COM_E)(all,r,REG)  

   qca(c,a,r) = qa(a,r); 

 

   zerodivide default 0.111; 

Coefficient (all,c,COM_E)(all,a,ACT)(all,r,REG)     SCOA(c,a,r) 

   # Share of commodity output by activity and region #; 

   Formula (all,c,COM_E)(all,a,ACT)(all,r,REG) 

   SCOA(c,a,r) = VCAM(c,a,r) / sum{p,COM_E, VCAM(p,a,r)} ; 

   Write SCOA to file OUTFILE; 

   zerodivide off; 

 

Equation DEF_PAM     

   # Unitary market revenue by activity and region # 

   (all,a,ACT)(all,r,REG)  

   pam(a,r) = sum{c,COM_E, SCOA(c,a,r) * pc(c,r)}; 

 

!--------------------------------------! 

!             6.1.5 STOCKS             ! 

!--------------------------------------! 

Coefficient (all,c,COM_E)(all,r,REG)     LEVPC(c,r) 

   # Level price of domestically-produced commodities by region #; 

   Formula (initial)(all,c,COM_E)(all,r,REG) LEVPC(c,r) = 1; 

   Update (all,c,COM_E)(all,r,REG) LEVPC(c,r) = pc(c,r); 

 

Equation DEF_DELQK 

   # Change in the volume of stock by commodity and region # 

   (all,c,COM_P)(all,r,REG) 

   100*LEVPC(c,r)*delqk(c,r) =  

   VKC(c,r)*sum[a,ACT, VCAM(c,a,r)*qca(c,a,r)] + stockslack(c,r); 

!  The standard closure specifies stockslack as exogenous and delqk as endogenous 

   variables; this implies that the change in the volume of stock of commodity c 

   changes proportionally to changes in aggregate production of commodity c. 

   Alternatively, by swapping stockslack by delqk, then changes in stocks are 

   exogenize and consequently subject to be shocked by the user! 

 

Equation DEF_DELVK 

   # Change in the value of stock by commodity and region # 

   (all,c,COM_P)(all,r,REG) 

   delvk(c,r) = 0.01 * VKC(c,r) * pc(c,r) + LEVPC(c,r) * delqk(c,r); 

!  the form of this equation is derived as follows: 

   VKC = QK*PC; delvk = QK*delpc + PC*delqk; delvk = QK*(PC*pc/100) + PC * delqk; 

   Rearranging we obtain the right form: delvk = 0.01 * VKC * pc + PC * delqk. 

   See that we need the data on the level of price pc. We arbitrarily set PC = 1 

   in the Coefficient statement above. ! 

!-------------------------------------! 

!   6.1.6 MARKETED COMMODITY OUTPUT   ! 

!-------------------------------------! 

Coefficient(all,c,COM_E)(all,r,REG)   VTOC(c,r) 

   # Value of traded output by commodity and region #; 

   Formula(all,c,COM_P)(all,r,REG) 

   VTOC(c,r) = sum{a,ACT, VCAM(c,a,r)} - VKC(c,r); 

   Formula(all,c,COM_ECP)(all,r,REG) 

   VTOC(c,r) = sum{a,ACT, VCAM(c,a,r)}; 

   Write VTOC to file OUTFILE; 

 

   zerodivide default 0.111; 

Coefficient (all,c,COM_E)(all,a,ACT)(all,r,REG)     SCATCO(c,a,r) 

   # Share of comm by activity in total comm output by region #; 

   Formula (all,c,COM_E)(all,a,ACT)(all,r,REG) 

   SCATCO(c,a,r) = VCAM(c,a,r) / sum{s,ACT, VCAM(c,s,r)} ; 

   Write SCATCO to file OUTFILE; 

   zerodivide off; 

 

Equation DEF_QC_P    

   # Volume of commercialization by commodity and region # 

   (all,c,COM_P)(all,r,REG)  

   [VTOC(c,r) + TINY] * qc(c,r) = sum[a,ACT, VCAM(c,a,r) * qca(c,a,r)] -  

                                  100 * LEVPC(c,r) * delqk(c,r); 

!  In levels, this clearing equation is qc = SUM{ACT, QCA} - DELQK. In producer 

   values, PC*qc = SUM{ACT, PC*QCA} - PC*DELQK ==> VTOC = SUM{ACT, VCAM} - VKC.   

   In percentage-change, it becomes VTOC*qc = SUM{ACT, VCAM*qca} - VKC*qk. 

   However, there is no percentage-change variable qk, but rather the change  

   variable delqk. The relationship between delqk and qk is qk = delqk/DELQK*100. 

   Replacing we get VTOC*qc = SUM{ACT, VCAM*qca} - VKC*(delqk/DELQK*100). 

   Simplifying we finally get VTOC*qc = SUM{ACT, VCAM*qca} - 100*LEVPC*delqk.! 

 

Equation DEF_QC_ECP    

   # Volume of commercialization by commodity and region # 

   (all,c,COM_ECP)(all,r,REG)  

   qc(c,r) = sum{a,ACT, SCATCO(c,a,r) * qca(c,a,r)}; 

!--------------------------------------! 

!    6.2 EXOGENOUS COMMODITY SUPPLY    ! 

!--------------------------------------! 

Equation DEF_QZ   

   # Supply by exogenous commodity and region # 

   (all,c,COM_CE)(all,r,REG)  

   qz(c,r) = qzint(c,r) + ESUPC(c,r) * pz(c,r); 

!  A perfectly inelastic supply function can be specified by choosing very small 

   values for ESUBC in the database rather than eliminating the equation above  

   and exogenizing qz(c,r)! 

 

!--------------------------------------! 

!          6.3 FACTOR SUPPLY           ! 

!--------------------------------------! 

Equation DEF_QFS    

   # Total supply by factor of production and region # 

   (all,f,FAC)(all,r,REG)  

   qfs(f,r) = qfsint(f,r) + ESUPF(f,r) * pf(f,r); 

!  A perfectly inelastic supply function can be specified by choosing very small 

   values for ESUBF in the database rather than eliminating the equation above  

   and exogenizing qfs(f,r)! 

 

Equation DEF_QFSA    

   # Sectoral supply by sluggish factor of production and region # 

   (all,f,FS)(all,a,ACT)(all,r,REG)  

   qfsa(f,a,r) = qfs(f,r) + ETRAF(f,r) * [pf(f,r) - pfas(f,a,r)]; 

 

   zerodivide default 0.111; 

Coefficient (all,f,FS)(all,a,ACT)(all,r,REG)     SATFS(f,a,r) 

   # Share of sectoral to total endowment of sluggish factors by region #; 

   Formula (all,f,FS)(all,a,ACT)(all,r,REG) 

   SATFS(f,a,r) = VEA(f,a,r) / sum{s,ACT, VEA(f,s,r)} ; 

   Write SATFS to file OUTFILE; 

   zerodivide off; 

 

Equation DEF_PF     

   # Price of factors of production by region # 

   (all,f,FS)(all,r,REG)  

   pf(f,r) = sum{a,ACT, SATFS(f,a,r) * pfas(f,a,r)}; 

!--------------------------------------! 

!        6.4 IMPORT ALLOCATION         ! 

!--------------------------------------! 

Equation DEF_QMS     

   # Bilateral volume of imports by commodity # 

   (all,c,COM_E)(all,r,REG)(all,s,REG)  

   qms(c,r,s) = qmsint(c,r,s) + qm(c,s)+ ESUBM(c,s)*[pmm(c,s) - pmms(c,r,s)]; 

 

   zerodivide default 0.0116; 

Coefficient (all,c,COM_E)(all,r,REG)(all,s,REG)     SBTYC(c,r,s) 

   # Share of bilateral to total imports by commodity by region #; 

   Formula (all,c,COM_E)(all,r,REG)(all,s,REG) 

   SBTYC(c,r,s) = VBYM(c,r,s) / sum{p,REG, VBYM(c,p,s)} ; 

   Write SBTYC to file OUTFILE; 

   zerodivide off; 

 

Equation DEF_PMM     

   # Market price of imports by commodity and region # 

   (all,c,COM_E)(all,r,REG)  

   pmm(c,r) = sum{s,REG, SBTYC(c,s,r) * pmms(c,s,r)}; 

 

!--------------------------------------! 

!       6.5 COMPOSITE COMMODITY        ! 

!--------------------------------------! 

Equation DEF_QD     

   # Volume of domestic commodity traded domestically by region # 

   (all,c,COM_E)(all,r,REG)  

   qd(c,r) = qq(c,r) + ESUBQ(c,r) * [pq(c,r) - pc(c,r)]; 

 

Equation DEF_QM     

   # Volume of imports by commodity and region # 

   (all,c,COM_E)(all,r,REG)  

   qm(c,r) = qq(c,r) + ESUBQ(c,r) * [pq(c,r) - pmm(c,r)]; 

!--------------------------------------! 

!        6.6 FINAL CONSUMPTION         ! 

!--------------------------------------! 

!--------------------------------------! 

!         CDE functional form          ! 

!--------------------------------------! 

   zerodivide default 0.333; 

Coefficient(all,c,COM_FC)(all,r,REG)   SFCC(c,r) 

   # Share of commodity to total expenditure for final consumption by region #; 

   Formula(all,c,COM_FC)(all,r,REG) 

   SFCC(c,r) = VFC(c,r) / sum{p,COM_FC, VFC(p,r)}; 

   Write SFCC to file OUTFILE; 

   zerodivide off; 

 

Coefficient (all,c,COM_FC)(all,r,REG)   ALPHA(c,r) 

   # CDE substitution parameter #; 

   Formula (all,c,COM_FC)(all,r,REG) 

   ALPHA(c,r) = (1 - SUBPAR(c,r)); 

 

Coefficient (all,c,COM_FC)(all,g,COM_FC)(all,r,REG)             APE(c,g,r) 

   # Allen partial elasticity of subst. btwn rice types c and g by region #; 

   Formula (all,c,COM_FC)(all,g,COM_FC)(all,r,REG) 

   APE(c,g,r) = ALPHA(c,r) + ALPHA(g,r)- sum[p,COM_FC, SFCC(p,r) * ALPHA(p,r)]; 

 

Formula (all,c,COM_FC)(all,r,REG) 

   APE(c,c,r) = 2 * ALPHA(c,r) - sum[p,COM_FC, SFCC(p,r) * ALPHA(p,r)] -  

                ALPHA(c,r) / [SFCC(c,r) + TINY]; 

 

Coefficient (all,c,COM_FC)(all,r,REG)     EY(c,r) 

   # income elasticity of demand for tradable commodity c by region #; 

   Formula (all,c,COM_FC)(all,r,REG) 

   EY(c,r) = 1 / sum[p,COM_FC, SFCC(p,r) * INCPAR(p,r)] *  

             {INCPAR(c,r) * [1 - ALPHA(c,r)]  

             + sum[p,COM_FC, SFCC(p,r) * INCPAR(p,r) * ALPHA(p,r)]} 

             + {ALPHA(c,r) - sum[p,COM_FC, SFCC(p,r) * ALPHA(p,r)]} ; 

Coefficient (all,c,COM_FC)(all,g,COM_FC)(all,r,REG)             EP(c,g,r) 

   # Uncompensated demand elasticity by commodity and region #; 

   Formula (all,c,COM_FC)(all,g,COM_FC)(all,r,REG) 

   EP(c,g,r) = 0;  

   Formula (all,c,COM_FC)(all,g,COM_FC)(all,r,REG) 

   EP(c,g,r) = [APE(c,g,r) - EY(c,r)] * SFCC(g,r);  

 

Equation DEF_QDFC 

   # Demand for final consumption by composite commodity and region # 

   (all,c,COM_FC)(all,r,REG) 

   qdfc(c,r) - pop(r) = sum[g,COM_FC, EP(c,g,r) * pr(g,r)] +  

                              EY(c,r) * [expn(r) - pop(r)]; 

 

!==============================================================================! 

!                         7. PRICE LINKAGE EQUATIONS                           ! 

!==============================================================================! 

Equation PL_ACTIVITY     

   # linkage between producer and market price by activity and region # 

   (all,a,ACT)(all,r,REG)  

   pap(a,r) = pam(a,r) + to(a,r); 

!  to(a) is exogenous in the standard model, and represents all domestic support 

   or taxes that create a wedge between the price producers received from the  

   market and the final price including government intervention! 

 

Equation PL_FM     

   # linkage between the market and activity price of factors by region # 

   (all,f,FM)(all,a,ACT)(all,r,REG)  

   pfa(f,a,r) = pf(f,r) + tfa(f,a,r); 

!  tfa(f,a,r) is exogenous in the standard model, and represents the power of  

   taxes/subsidies on factors of production. tfa(f,a,r) creates a wedge between the 

   factor market price and what producers actually pay for their use! 

 

Equation PL_FS     

   # linkage between the market and activity price of factors by region # 

   (all,f,FS)(all,a,ACT)(all,r,REG)  

   pfa(f,a,r) = pfas(f,a,r) + tfa(f,a,r); 

 

Equation PL_INPUT     

   # linkage between the market and activity price of exogenous inputs by region # 

   (all,c,COM_CE)(all,a,ACT)(all,r,REG)  

   pza(c,a,r) = pz(c,r) + tia(c,a,r); 

Equation PL_EXP1     

   # Linkage between market and world price of bilateral exports by commodity  # 

   (all,c,COM_E)(all,r,REG)(all,s,REG)  

   pxbd(c,r,s) = pc(c,r) + xmrg(c,r,s); 

 

Equation PL_EXP2     

   # Linkage between market and world price of bilateral exports by commodity  # 

   (all,c,COM_E)(all,r,REG)(all,s,REG)  

   pxwd(c,r,s) = pxbd(c,r,s) + txd(c,r,s); 

 

   zerodivide default 0.0116; 

Coefficient(all,c,COM_E)(all,r,REG)(all,s,REG)   SFOBCIF(c,r,s) 

   # Share of fob to cif value of imports by region #; 

   Formula(all,c,COM_E)(all,r,REG)(all,s,REG) 

   SFOBCIF(c,r,s) = VBXW(c,r,s) / [VBYW(c,r,s)]; 

   Write SFOBCIF to file OUTFILE; 

   zerodivide off; 

 

Equation PL_FOBCIF     

   # Linkage between world price of bilateral exports and imports by commodity # 

   (all,c,COM_E)(all,r,REG)(all,s,REG)  

   pmws(c,r,s) = SFOBCIF(c,r,s) * pxwd(c,r,s) + [1-SFOBCIF(c,r,s)] * tc(c,r,s); 

 

Equation PL_IMP     

   # Linkage between market and world price of bilateral imports by commodity # 

   (all,c,COM_E)(all,r,REG)(all,s,REG)  

   pmms(c,r,s) = pmws(c,r,s) + tms(c,r,s); 

 

Equation PL_FC1     

   # Linkage between bulk and wholesale price of final consumption # 

   (all,c,COM_FC)(all,r,REG)  

   pwh(c,r) = pq(c,r) + wmrg(c,r); 

 

Equation PL_FC2     

   # Linkage between wholesale and retail price of final consumption # 

   (all,c,COM_FC)(all,r,REG)  

   pr(c,r) = pwh(c,r) + rmrg(c,r); 

!==============================================================================! 

!                             8. SYSTEM CONSTRAINTS                            ! 

!==============================================================================! 

!--------------------------------------! 

!      8.1 ZERO PROFIT CONDITIONS      ! 

!--------------------------------------! 

! Zero profit conditions determine sector's output, or which is the same, output 

  is the free variable that clears the zero-profit condition for each sector ! 

 

   zerodivide default 0.111; 

Coefficient(all,f,FAC)(all,a,ACT)(all,r,REG)   SFTAC(f,a,r) 

   # Share of factor in total activity cost by region #; 

   Formula(all,f,FAC)(all,a,ACT)(all,r,REG) 

   SFTAC(f,a,r) = VEA(f,a,r) /  

                      [sum{g,FAC,VEA(g,a,r)} + sum{c,COM_CFC,VICA(c,a,r)}]; 

   Write SFTAC to file OUTFILE; 

 

Coefficient(all,c,COM_CFC)(all,a,ACT)(all,r,REG)   SITAC(c,a,r) 

   # Share of input in total activity cost by region #; 

   Formula(all,c,COM_CFC)(all,a,ACT)(all,r,REG) 

   SITAC(c,a,r) = VICA(c,a,r) /  

                      [sum{f,FAC,VEA(f,a,r)} + sum{p,COM_CFC,VICA(p,a,r)}]; 

   Write SITAC to file OUTFILE; 

 

Coefficient(all,c,COM_E)(all,r,REG)   VOQ(c,r) 

   # Output value of composite commodity c by region #; 

   Formula(all,c,COM_E)(all,r,REG) 

   VOQ(c,r) = VTOC(c,r) - sum{s,REG, VBXM(c,r,s)} + sum{s,REG, VBYM(c,s,r)}; 

 

Coefficient(all,c,COM_E)(all,r,REG)   SYQO(c,r) 

   # Share of imports in total value of composite commodity c by region #; 

   Formula(all,c,COM_E)(all,r,REG) 

   SYQO(c,r) = sum{s,REG, VBYM(c,s,r)} / VOQ(c,r); 

   Write SYQO to file OUTFILE; 

   zerodivide off; 

Equation ZP_PRODUCTION     

   # zero profits condition by activity and region # 

   (all,a,ACT)(all,r,REG)  

   [pap(a,r)+ao(a,r)] =  

          sum{f,FAC, SFTAC(f,a,r) * [pfa(f,a,r)-afe(f,a,r)-ava(a,r)]} +  

          sum{c,COM_ECFC, SITAC(c,a,r) * [pq(c,r)-aie(c,a,r)-ain(a,r)]} + 

          sum{c,COM_CE, SITAC(c,a,r) * [pza(c,a,r)-aie(c,a,r)-ain(a,r)]} + 

          zpactslack(a,r); 

!  In the standard closure, "zpactslack(a,r)" is treated as exogenous and qa(a,r) 

   is the endogenous free variable that clears this accounting condition. 

   Alternatively, if the user wants to fix the activity level qa(a,r) for some 

   particular sector, then he/she should swap zpactslack(a,r) for qa(a,r). In 

   this closure, "zpactslack(a,r)" will indicate the percentage change in profits 

   by activity and region! 

!  At equilibrium, this equation determines the unitary cost by activity pap(a,r)! 

 

Equation ZP_COMPOSITE     

   # zero profits condition in the production of composite commodity c  # 

   (all,c,COM_E)(all,r,REG)  

   pq(c,r) = SYQO(c,r) * pmm(c,r) + [1 - SYQO(c,r)] * pc(c,r); 

!  qq(c,r) is the free variable that balances this account! 

!  At equilibrium, this equation determines the composite commodity price pq(c,r)! 

 

!--------------------------------------! 

!    8.2 MARKET CLEARING CONDITIONS    ! 

!--------------------------------------! 

!Market clearing conditions determine prices, or which is the same, price is  

 the free variable that clears each relevant market (commodities and factors)! 

 

Equation MKTCLR_FC    

   # market clearing in final consumption of composite commodity c by region # 

   (all,c,COM_FC)(all,r,REG)  

   qq(c,r) = qdfc(c,r); 

!  pq(c,r) is the free variable that adjusts to clear this accounting condition ! 

 

Coefficient (all,c,COM_ECFC)(all,a,ACT)(all,r,REG)     SAIN(c,a,r) 

   # Share of activity to total use of inputs by region #; 

   Formula (all,c,COM_ECFC)(all,a,ACT)(all,r,REG) 

   SAIN(c,a,r) = VICA(c,a,r) / sum{s,ACT, VICA(c,s,r)} ; 

   Write SAIN to file OUTFILE; 

Equation MKTCLR_INTERM    

   # market clearing in intermediate consumption of composite commodity # 

   (all,c,COM_ECFC)(all,r,REG)  

   qq(c,r) = sum{a,ACT, SAIN(c,a,r) * qin(c,a,r)}; 

!  pq(c,r) is the free variable that adjusts to clear this accounting condition ! 

 

   zerodivide default 0.0116; 

Coefficient(all,c,COM_E)(all,r,REG)(all,s,REG)   SBXDO(c,r,s) 

   # Share of bilateral imports in total traded value of domestic comm c #; 

   Formula(all,c,COM_E)(all,r,REG)(all,s,REG) 

   SBXDO(c,r,s) = VBXM(c,r,s) / VTOC(c,r); 

   Write SBXDO to file OUTFILE; 

   zerodivide off; 

 

Equation MKTCLR_DOMCOM    

   # market clearing for domestic commodities by region # 

   (all,c,COM_E)(all,r,REG)  

   qc(c,r) = sum{s,REG, SBXDO(c,r,s) * qms(c,r,s)} +  

             [1 - sum{s,REG,SBXDO(c,r,s)}] * qd(c,r); 

!  pc(c,r) is the free variable that adjusts to clear this accounting condition ! 

 

   zerodivide default 0.111; 

Coefficient(all,c,COM_CE)(all,a,ACT)(all,r,REG)   SAZ(c,a,r) 

   # Share of activity to total use of exogenous commodity #; 

   Formula(all,c,COM_CE)(all,a,ACT)(all,r,REG) 

   SAZ(c,a,r) = VICA(c,a,r) / sum{s,ACT, VICA(c,s,r)}; 

   Write SAZ to file OUTFILE; 

   zerodivide off; 

 

Equation MKTCLR_EXOCOM     

   # market clearing for exogenous commodities by region # 

   (all,c,COM_CE)(all,r,REG)  

   qz(c,r) = sum{a,ACT, SAZ(c,a,r) * qin(c,a,r)} + excomslack(c,r); 

!  either pz(c,r) or excomslack(c,r) are the free variable that adjust to clear 

   this accounting condition depending on the closure chosen.  

   If the user wants to exogenize the commodity price pz(c,r) for exogenous 

   commodities, then excomslack(c,r) must be set endogenous and pz(c,r) exogenous. 

   Otherwise, if the user wants the market-clearing condition to prevail, then 

   pz(c,r) must be set endogenous and excomslack(c,r) exogenous. the later is the 

   default closure ! 

Equation MKTCLR_FM     

   # clearing in the market for mobile factors by region # 

   (all,f,FM)(all,r,REG)  

   [sum{a,ACT, VEA(f,a,r)} + TINY] * qfs(f,r) =  

                                           sum{a,ACT, VEA(f,a,r) * qfa(f,a,r)}; 

!  pf(f,r) is the free variable that adjusts to clear this accounting condition ! 

 

Equation MKTCLR_FS     

   # clearing in the market for sluggish factors by region # 

   (all,a,ACT)(all,f,FS)(all,r,REG)  

   qfsa(f,a,r) = qfa(f,a,r); 

!  the activity-specific factor price pfas(f,a,r) is the free variable that  

   adjusts so as to clear the market for each sluggish factor ! 

 

!==============================================================================! 

!                          9. SECONDARY ESTIMATIONS                            ! 

!==============================================================================! 

!------------------------------! 

!  9.1 PRODUCTION AGGREGATES   ! 

!------------------------------! 

!---------------------------------------------------! 

!  9.1.1 TOTAL PRODUCTION OF PADDY RICE BY REGION   ! 

!---------------------------------------------------! 

Variable 

(all,r,REG)   qpr(r) # total production of paddy rice by region #; 

 

Equation DEF_QPR     

   # regional production of paddy rice # 

   (all,r,REG)  

   [sum{c,COM_P, sum{a,ACT, VCAM(c,a,r)}} + TINY] * qpr(r) =  

                             sum{c,COM_P, sum{a,ACT, VCAM(c,a,r) * qca(c,a,r)}}; 

 

!---------------------------------------------------! 

!  9.1.2 TOTAL PRODUCTION OF BROWN RICE BY REGION   ! 

!---------------------------------------------------! 

!This is an indicator of the change in the initial milling activity in a region! 

SET COM_B  # brown rice commodities # (C_LGB,C_MGB,C_FRB); 

           SUBSET COM_B IS SUBSET OF COM_E; 

Variable 

(all,r,REG)   qbr(r) # total production of brown rice by region #; 

Equation DEF_QBR     

   # regional production of brown rice # 

   (all,r,REG)  

   [sum{c,COM_B, sum{a,ACT, VCAM(c,a,r)}} + TINY] * qbr(r) =  

                             sum{c,COM_B, sum{a,ACT, VCAM(c,a,r) * qca(c,a,r)}}; 

 

!---------------------------------------------------! 

!  9.1.3 TOTAL PRODUCTION OF WHITE RICE BY REGION   ! 

!---------------------------------------------------! 

!This is an indicator of the change in the final milling activity in a region! 

Variable 

(all,r,REG)   qwr(r) # total production of white rice by region #; 

 

Equation DEF_QWR     

   # regional production of white rice # 

   (all,r,REG)  

   [sum{c,COM_FC, sum{a,ACT, VCAM(c,a,r)}} + TINY] * qwr(r) =  

                            sum{c,COM_FC, sum{a,ACT, VCAM(c,a,r) * qca(c,a,r)}}; 

 

!-----------------------------------------------------! 

!  9.1.4 WORLDWIDE PRODUCTION OF PADDY RICE BY TYPE   ! 

!-----------------------------------------------------! 

Variable 

(all,c,COM_P)   qpt(c) # worldwide production of paddy rice by type #; 

 

Equation DEF_QPT     

   # worldwide production of paddy rice by type # 

   (all,c,COM_P)  

   [sum{a,ACT, sum{r,REG, VCAM(c,a,r)}} + TINY] * qpt(c) =  

                               sum{a,ACT, sum{r,REG, VCAM(c,a,r) * qca(c,a,r)}}; 

 

!---------------------------------------------! 

!  9.1.5 WORLDWIDE PRODUCTION OF PADDY RICE   ! 

!---------------------------------------------! 

Variable 

                qp     # worldwide production of paddy rice #; 

 

Equation DEF_QP     

   # worldwide production of paddy rice # 

   [sum{c,COM_P, sum{a,ACT, sum{r,REG, VCAM(c,a,r)}}} + TINY] * qp =  

                 sum{c,COM_P, sum{a,ACT, sum{r,REG, VCAM(c,a,r) * qca(c,a,r)}}}; 

!-------------------------------! 

!  9.2 CONSUMPTION AGGREGATES   ! 

!-------------------------------! 

!-----------------------------------------------! 

!  9.2.1 TOTAL CONSUMPTION OF RICE BY REGION    ! 

!-----------------------------------------------! 

Variable 

(all,r,REG)   qfcr(r) # total consumption of rice by region #; 

 

Equation DEF_QFCR     

   # total consumption of rice by region # 

   (all,r,REG)  

   [sum{c,COM_FC, VFC(c,r)} + TINY] * qfcr(r) = sum{c,COM_FC,VFC(c,r) * qdfc(c,r)}; 

 

!-------------------------------------------------! 

!  9.2.2 WORLDWIDE CONSUMPTION OF RICE BY TYPE    ! 

!-------------------------------------------------! 

Variable 

(all,c,COM_FC)   qfct(c) # worldwide final consumption of rice by type #; 

 

Equation DEF_QFCT     

   # worldwide final consumption of rice by type # 

   (all,c,COM_FC)  

   [sum{r,REG, VFC(c,r)} + TINY] * qfct(c) = sum{r,REG, VFC(c,r) * qdfc(c,r)}; 

 

!-----------------------------------------! 

!  9.2.3 WORLDWIDE CONSUMPTION OF RICE    ! 

!-----------------------------------------! 

Variable 

                 qfc # worldwide consumption of rice #; 

 

Equation DEF_QFC     

   # worldwide consumption of rice # 

   [sum{r,REG, sum{c,COM_FC, VFC(c,r)}} + TINY] * qfc =  

                                sum{r,REG, sum{c,COM_FC, VFC(c,r) * qdfc(c,r)}}; 

!-------------------------! 

!  9.3 TRADE AGGREGATES   ! 

!-------------------------! 

!----------------------------------------------------! 

!  9.3.1 AGGREGATE EXPORTS BY COMMODITY AND REGION   ! 

!----------------------------------------------------! 

Variable  

(all,c,COM_E)(all,r,REG)    pxcr(c,r) # export price index by comm and region #; 

(all,c,COM_E)(all,r,REG)    qxcr(c,r) # volume of export by comm and region #; 

 

Equation DEF_PXCR     

   # world export price by comm and region # 

   (all,c,COM_E)(all,r,REG)  

   [sum{s,REG, VBXW(c,r,s)} + TINY] * pxcr(c,r) =  

                                             sum{s,REG, VBXW(c,r,s) * pxwd(c,r,s)}; 

 

Equation DEF_QXCR     

   # volume of export by comm and region # 

   (all,c,COM_E)(all,r,REG)  

   [sum{s,REG, VBXW(c,r,s)} + TINY] * qxcr(c,r) =  

                                          sum{s,REG, VBXW(c,r,s) * qms(c,r,s)}; 

 

!--------------------------------------! 

!  9.3.2 AGGREGATE EXPORTS BY REGION   ! 

!--------------------------------------! 

Variable  

(all,r,REG)    qxr(r) # volume of export by region #; 

 

Equation DEF_QXR     

   # volume of import by region # 

   (all,r,REG)  

   [sum{s,REG,sum{c,COM_E, VBYW(c,r,s)}} + TINY] * qxr(r) =  

                           sum{s,REG,sum{c,COM_E, VBYW(c,r,s) * qms(c,r,s)}}; 

 

!-----------------------------------------------------! 

!  9.3.3 AGGREGATE IMPORTS BY COMMODITY AND COUNTRY   ! 

!-----------------------------------------------------! 

Variable  

(all,c,COM_E)(all,r,REG)    pmcr(c,r) # import price index by comm and region #; 

(all,c,COM_E)(all,r,REG)    qmcr(c,r) # volume of import by comm and region #; 

Equation DEF_PMCR     

   # world import price by comm and region # 

   (all,c,COM_E)(all,r,REG)  

   [sum{s,REG, VBYW(c,s,r)} + TINY] * pmcr(c,r) =  

                                          sum{s,REG, VBYW(c,s,r) * pmws(c,s,r)}; 

 

Equation DEF_QMCR     

   # volume of import by comm and region # 

   (all,c,COM_E)(all,r,REG)  

   [sum{s,REG, VBYW(c,s,r)} + TINY] * qmcr(c,r) =  

                                             sum{s,REG, VBYW(c,s,r) * qms(c,s,r)}; 

 

!---------------------------------------! 

!  9.3.4 AGGREGATE IMPORTS BY COUNTRY   ! 

!---------------------------------------! 

Variable  

(all,r,REG)    qmr(r) # volume of import by region #; 

 

Equation DEF_QMR     

   # volume of import by region # 

   (all,r,REG)  

   [sum{s,REG,sum{c,COM_E, VBYW(c,s,r)}} + TINY] * qmr(r) =  

                              sum{s,REG,sum{c,COM_E, VBYW(c,s,r) * qms(c,s,r)}}; 

 

!-------------------------------------! 

!  9.3.5 TOTAL BILATERAL RICE TRADE   ! 

!-------------------------------------! 

Variable  

(all,r,REG)(all,s,REG)    qmst(s,r) # total volume of bilateral trade #; 

 

Equation DEF_QMST     

   # total volume of bilateral trade # 

   (all,s,REG)(all,r,REG)  

   [sum{c,COM_E, VBXW(c,s,r)} + TINY] * qmst(s,r) =  

                                         sum{c,COM_E, VBXW(c,s,r) * qms(c,s,r)}; 

 

!------------------------------------------------! 

!  9.4 WORLD PRICE & TRADE VOLUME BY COMMODITY   ! 

!------------------------------------------------! 

Variable  

(all,c,COM_E)     pcw(c)  # world (f.o.b.) price index by commodity #; 

(all,c,COM_E)     qcwt(c) # Volume of world trade by commodity #; 

Equation DEF_PCW     

   # World (f.o.b.) price index by commodity # 

   (all,c,COM_E)  

   [sum{r,REG, sum{s,REG, VBXW(c,r,s)}} + TINY] * pcw(c) =  

                              sum{r,REG, sum{s,REG, VBXW(c,r,s) * pxwd(c,r,s)}}; 

 

Equation DEF_QCWT     

   # Volume of world trade by commodity # 

   (all,c,COM_E)  

   [sum{r,REG, sum{s,REG, VBXW(c,r,s)}} + TINY] * qcwt(c) = 

                              sum{r,REG, sum{s,REG, VBXW(c,r,s) * qms(c,r,s)}}; 

 

!--------------------------------------! 

!  9.5 RICE WORLD PRICE & TRADE INDEX  ! 

!--------------------------------------! 

Variable        pw  # Rice world price index #; 

                qwt # Rice world trade index #; 

 

Equation DEF_PW     

   # Rice world price index # 

   [sum{c,COM_E, sum{r,REG, sum{s,REG, VBXW(c,r,s)}}} + TINY] * pw = 

               sum{c,COM_E, sum{r,REG, sum{s,REG, VBXW(c,r,s) * pxwd(c,r,s)}}}; 

 

Equation DEF_QWT     

   # Rice world trade index # 

   [sum{c,COM_E, sum{r,REG, sum{s,REG, VBXW(c,r,s)}}} + TINY] * qwt = 

               sum{c,COM_E, sum{r,REG, sum{s,REG, VBXW(c,r,s) * qms(c,r,s)}}}; 

 

 


